Lesson 25: Solving equations using series

```
> restart;
```

Example 2: A series for an implicit function

Find the Taylor series for $y(x)$ about $x = 0$ (up to the x^6 term), if $y = y(x)$ satisfies the equation $(1 + x)\, e^y - y^2\, e^x = 1 + x^2\, y$ with $y(0) = 0$.

Last time we saw that a version of Newton's method can be used.

```
> eq:= (1+x)*exp(y)-y^2*exp(x) = 1 + x^2*y;
```

$$eq := (1 + x)\, e^y - y^2\, e^x = 1 + x^2\, y \qquad \textbf{(1.1)}$$

```
> f:= unapply(lhs(eq)-rhs(eq),(x,y));
```

$$f := (x, y) \rightarrow (1 + x)\, e^y - y^2\, e^x - 1 - x^2\, y$$

```
> newt:= (y,n) -> convert(normal(taylor(y-f(x,y)/D[2](f)(x,y),
  x, n)),polynom);
```

$$newt := (y, n) \rightarrow convert\left(normal\left(taylor\left(y - \frac{f(x, y)}{D_2(f)\,(x, y)}, x, n\right)\right), polynom\right)$$

```
> y1 := newt(0,2);
```

$$y1 := -x$$

```
> y2 := newt(y1,4);
```

$$y2 := -x + \frac{3}{2}\, x^2 - \frac{10}{3}\, x^3$$

If $f(x, y_k) = O(x^k)$ and $\frac{\partial}{\partial y} f(x, y_k)$ has a nonzero limit as $x \rightarrow 0$, then $f(x, y_{k+1}) = O(x^{2k})$ where $y_{k+1} = y_k - \dfrac{f(x, y_k)}{\frac{\partial}{\partial y} f(x, y_k)}$.

In other words, once you get an approximation that works to a certain order $O(x^k)$, each application of Newton's method will at least double the order of approximation.

```
> y3 := newt(y2,8);
```

$$y3 := -x + \frac{3}{2}\, x^2 - \frac{10}{3}\, x^3 + \frac{23}{3}\, x^4 - \frac{1097}{60}\, x^5 + \frac{8117}{180}\, x^6 - \frac{285673}{2520}\, x^7$$

Notice that the terms in x, x^2, x^3 are the same as in y2.

```
> y4 := newt(y3, 16);
```

$$y4 := -x + \frac{3}{2}\, x^2 - \frac{10}{3}\, x^3 + \frac{23}{3}\, x^4 - \frac{1097}{60}\, x^5 + \frac{8117}{180}\, x^6 - \frac{285673}{2520}\, x^7 + \frac{242153}{840}\, x^8 - \frac{1061687}{1440}\, x^9 + \frac{1141506817}{604800}\, x^{10} - \frac{48062135701}{9979200}\, x^{11} + \frac{974043196177}{79833600}\, x^{12} - \frac{95110784043697}{3113510400}\, x^{13} + \frac{546125185861933}{7264857600}\, x^{14} - \frac{16853821021600523}{93405312000}\, x^{15}$$

I'll switch to using floating-point (by sticking in an **evalf**), because some of these coefficients are starting to involve rational numbers with big numerators and denominators.

```
> y5 := newt(evalf(y4), 32);
```

$$y5 := -1.\, x + 1.500000000\, x^2 - 3.333333333\, x^3 + 7.666666663\, x^4 - 18.28333333\, x^5 + 45.09444446\, x^6 - 113.3623015\, x^7 + 288.2773811\, x^8 - 737.2826390\, x^9 + 1887.412065\, x^{10} - 4816.231328\, x^{11} + 12200.91789\, x^{12} - 30547.76497\, x^{13} + 75173.55676\, x^{14} - 1.804374995\; 10^5\, x^{15} + 4.174699534\; 10^5\, x^{16} - 9.118492738\; 10^5\, x^{17} + 1.800051234\; 10^6\, x^{18} - 2.838448116\; 10^6\, x^{19} + 1.536521088\; 10^6\, x^{20} + 1.426525873\; 10^7\, x^{21} - 9.335475933\; 10^7\, x^{22} + 4.138114467\; 10^8\, x^{23} - 1.589561380\; 10^9\, x^{24} + 5.655720661\; 10^9\, x^{25} - 1.916297530\; 10^{10}\, x^{26} + 6.270482445\; 10^{10}\, x^{27} - 1.997374001\; 10^{11}\, x^{28} + 6.223365828\; 10^{11}\, x^{29} - 1.902341622\; 10^{12}\, x^{30} + 5.715120029\; 10^{12}\, x^{31}$$

This polynomial might not be sorted in order of the exponents. We can use **sort** to fix this.

```
> sort(y5, x, ascending);
```

$$-1.\, x + 1.500000000\, x^2 - 3.333333333\, x^3 + 7.666666663\, x^4 - 18.28333333\, x^5 + 45.09444446\, x^6 - 113.3623015\, x^7 + 288.2773811\, x^8 - 737.2826390\, x^9 + 1887.412065\, x^{10} - 4816.231328\, x^{11} + 12200.91789\, x^{12} - 30547.76497\, x^{13} + 75173.55676\, x^{14} - 1.804374995\; 10^5\, x^{15} + 4.174699534\; 10^5\, x^{16} - 9.118492738\; 10^5\, x^{17} + 1.800051234\; 10^6\, x^{18} - 2.838448116\; 10^6\, x^{19} + 1.536521088\; 10^6\, x^{20} + 1.426525873\; 10^7\, x^{21} - 9.335475933\; 10^7\, x^{22} + 4.138114467\; 10^8\, x^{23} - 1.589561380\; 10^9\, x^{24} + 5.655720661\; 10^9\, x^{25} - 1.916297530\; 10^{10}\, x^{26} + 6.270482445\; 10^{10}\, x^{27} - 1.997374001\; 10^{11}\, x^{28} + 6.223365828\; 10^{11}\, x^{29} - 1.902341622\; 10^{12}\, x^{30} + 5.715120029\; 10^{12}\, x^{31} \quad \textbf{(1.2)}$$

```
> y6:= sort(newt(evalf(y5),64), x, ascending);
```

$$y6 := -1.\, x + 1.500000000\, x^2 - 3.333333333\, x^3 + 7.666666669\, x^4 - 18.28333332\, x^5 + 45.09444442\, x^6 - 113.3623018\, x^7 + 288.2773810\, x^8 - 737.2826395\, x^9 + 1887.412065\, x^{10} - 4816.231337\, x^{11} + 12200.91784\, x^{12} - 30547.76496\, x^{13} + 75173.55700\, x^{14} - 1.804375001\; 10^5\, x^{15} + 4.174699567\; 10^5\, x^{16} - 9.118492796\; 10^5\, x^{17} + 1.800051253\; 10^6\, x^{18} - 2.838448165\; 10^6\, x^{19} + 1.536521332\; 10^6\, x^{20} + 1.426525847\; 10^7\, x^{21} - 9.335475828\; 10^7\, x^{22} + 4.138114435\; 10^8\, x^{23} - 1.589561372\; 10^9\, x^{24} + 5.655720642\; 10^9\, x^{25} - 1.916297523\; 10^{10}\, x^{26} + 6.270482426\; 10^{10}\, x^{27} - 1.997374004\; 10^{11}\, x^{28} + 6.223365837\; 10^{11}\, x^{29} - 1.902341633\; 10^{12}\, x^{30} + 5.715120033\; 10^{12}\, x^{31} - 1.689062000\; 10^{13}\, x^{32} + 4.912063172\; 10^{13}\, x^{33} - 1.405151490\; 10^{14}\, x^{34} + 3.949808232\; 10^{14}\, x^{35} - 1.088982176\; 10^{15}\, x^{36} + 2.936073135\; 10^{15}\, x^{37} - 7.705055139\; 10^{15}\, x^{38} + 1.953187460\; 10^{16}\, x^{39} - 4.720377022\; 10^{16}\, x^{40} + 1.060602590\; 10^{17}\, x^{41} - 2.090715181\; 10^{17}\, x^{42} + 2.975312966\; 10^{17}\, x^{43} \quad \textbf{(1.3)}$$

$$+ 8.86495164\ 10^{16}\, x^{44} - 3.142658004\ 10^{18}\, x^{45} + 1.773084316\ 10^{19}\, x^{46}$$
$$- 7.675919233\ 10^{19}\, x^{47} + 2.959398805\ 10^{20}\, x^{48} - 1.066902120\ 10^{21}\, x^{49}$$
$$+ 3.676706220\ 10^{21}\, x^{50} - 1.225458658\ 10^{22}\, x^{51} + 3.977643050\ 10^{22}\, x^{52}$$
$$- 1.262644071\ 10^{23}\, x^{53} + 3.930219312\ 10^{23}\, x^{54} - 1.201512023\ 10^{24}\, x^{55}$$
$$+ 3.610529448\ 10^{24}\, x^{56} - 1.066640132\ 10^{25}\, x^{57} + 3.096497034\ 10^{25}\, x^{58}$$
$$- 8.823296399\ 10^{25}\, x^{59} + 2.462763065\ 10^{26}\, x^{60} - 6.711763781\ 10^{26}\, x^{61}$$
$$+ 1.776763274\ 10^{27}\, x^{62} - 4.530176727\ 10^{27}\, x^{63}$$

Here's an animation with polynomials of degrees up to 63, showing how well this converges to a solution.

```
> with(plots):
  P0:= implicitplot(f(x,y),x=-0.5 .. 0.5, y = -1 .. 2,colour=
  blue):
```

```
> for j from 1 to 63 do
    frame[j]:= display([P0,plot(convert(taylor(y6,x,j+1),
  polynom),x=-0.5..0.5)],title=('Degree'=j),view=[-0.5..0.5,-1.
  .2])
  end do:
  display([seq(frame[j],j=1..63)],insequence=true);
```



What's the radius of convergence? From the pictures, I'd guess it's a bit more than 0.3.

The theoretical result is that $\frac{1}{R} = \lim sup_{n\to\infty}\left(|a_n|^{\frac{1}{n}}\right)$.

That means (in the case where R is finite and nonzero) that for every $\varepsilon > 0$, $|a_n|^{\frac{1}{n}} < \frac{1}{R} + \varepsilon$ when n is sufficiently large, but there exist arbitrarily large n with $\frac{1}{R} - \varepsilon < |a_n|^{\frac{1}{n}}$.

```
> L := [seq([n,evalf(abs(coeff(y6,x,n))^(1/n))],n=1..63)];
```

L := [[1, 1.], [2, 1.224744871], [3, 1.493801582], [4, 1.663993576], [5, 1.788179350], [6, 1.886631900], [7, 1.965601714], [8, 2.029907840], [9, 2.082696566], [10, 2.126114614], [11, 2.161680855], [12, 2.190446794], [13, 2.213086651], [14, 2.229925554], [15, 2.240896441], [16, 2.245376002], [17, 2.241732093], [18, 2.225952159], [19, 2.185927936], [20, 2.038576107], [21, 2.191189727], [22, 2.302920416], [23, 2.369428385], [24, 2.417611730], [25, 2.455271473], [26, 2.485875326], [27, 2.511308319], [28, 2.532723823], [29, 2.550882936], [30, 2.566312960], [31, 2.579389615], [32, 2.590382663], [33, 2.599481830], [34, 2.606810757], [35, 2.612432205], [36, 2.616345063], [37, 2.618470909], [38, 2.618623410], [39, 2.616443631], [40, 2.611255797], [41, 2.601700992], [42, 2.584568061], [43, 2.548949460], [44, 2.427585099], [45, 2.576623997], [46, 2.620901135], [47, 2.649036125], [48, 2.669828754], [49, 2.686243636], [50, 2.699660664], [51, 2.710840063], [52, 2.720246052], [53, 2.728181285], [54, 2.734850837], [55, 2.740395455], [56, 2.744909635], [57, 2.748451222], [58, 2.751045480], [59, 2.752684614], [60, 2.753322247], [61, 2.752860521], [62, 2.751123929], [63, 2.747805286]]

```
> pointplot(L);
```



It looks plausible (with a little imagination) that the lim sup is around 3, which would correspond to a radius of about 0.3.
Here's a different (possibly better) way to do it: plot $\ln(|a_n|)$. The idea here is that if $R > r$, then $|a_n| < c\, r^{-n}$ for some constant c, so $\ln(|a_n|) < \ln(c) - n \ln(r)$. Thus in this plot, every point should be below a straight line with slope $-\ln(r)$. The radius of convergence R is e^{-m} where m is the minimum slope such that all points are below a line of slope m.

```
> L2 := [seq([n,evalf(ln(abs(coeff(y6,x,n))))],n=1..63)]:
```

```
> pointplot(L2);P1:= %:
```




```
> display([P1, plot([L2[1],[60,64]],colour=blue)]);
```




```
> slope:= (64-L2[1,2])/(60-L2[1,1]);
```

$$slope := 1.084745763 \quad (1.4)$$

```
> R:= exp(-slope);
```

$$R := 0.3379877040 \quad (1.5)$$

By the way, here's another way to get the series.

```
> solve(eq,y);
```

$$RootOf\left(-e^{\_Z} - e^{\_Z} x + 1 + \_Z^2 e^x + x^2 \_Z\right) \quad (1.6)$$

```
> taylor(%,x,10);
```

$$-x + \frac{3}{2} x^2 - \frac{10}{3} x^3 + \frac{23}{3} x^4 - \frac{1097}{60} x^5 + \frac{8117}{180} x^6 - \frac{285673}{2520} x^7 + \frac{242153}{840} x^8 - \frac{1061687}{1440} x^9 + O\left(x^{10}\right) \quad (1.7)$$

```
> taylor(%% - y4, x, 16);
```

$$O\left(x^{16}\right) \quad (1.8)$$

There's something slightly fishy about this:
where did I tell Maple that I wanted the solution with $y(0) = 0$?
I didn't. Now it just happens that the only real solution at $x = 0$ is $y = 0$. But for equations that don't

have that property, you might not know which solution you'll get by this method. For example, the next example has two solutions at $x = 0$: $y = 1$ and $y = -1$.

```
> eq2:= y^2 + x*exp(y) = 1;
```

$$eq2 := y^2 + e^y x = 1 \tag{1.9}$$

```
> solve(eq2,y);
```

$$RootOf(e^{\_Z} x + \_Z^2 - 1) \tag{1.10}$$

```
> taylor(%, x, 10);
```

$$-1 + \frac{1}{2} e^{-1} x + \frac{3}{8} e^{-2} x^2 + \frac{7}{16} e^{-3} x^3 + \frac{235}{384} e^{-4} x^4 + \frac{121}{128} e^{-5} x^5 + \frac{7959}{5120} e^{-6} x^6 + \frac{245953}{92160} e^{-7} x^7 + \frac{5422687}{1146880} e^{-8} x^8 + \frac{3936241}{458752} e^{-9} x^9 + O(x^{10}) \tag{1.11}$$

I get the solution with $y(0) = -1$, but I wouldn't have known this ahead of time. If I want the solution with $y(0) = 1$, I could put an extra argument on the RootOf that says what the solution should be at $x = 0$.

```
> taylor(RootOf(exp(_Z)*x + _Z^2 - 1, 1), x, 10);
```

$$1 - \frac{1}{2} e\, x + \frac{1}{8} e^2 x^2 - \frac{1}{16} e^3 x^3 + \frac{13}{384} e^4 x^4 - \frac{1}{48} e^5 x^5 + \frac{69}{5120} e^6 x^6 - \frac{841}{92160} e^7 x^7 + \frac{65689}{10321920} e^8 x^8 - \frac{10427}{2293760} e^9 x^9 + O(x^{10}) \tag{1.12}$$

From series to function

So far we've had a function and wanted to know its Taylor series. Now suppose you know the series but you want to identify the function. Maple might be able to do it with **sum**, if you know a formula for the coefficients.

```
> sum(k/(k+1)*x^k,k=0..infinity);
```

$$\frac{1}{2} x \left(-\frac{2}{x (x-1)} + \frac{2 \ln(1-x)}{x^2} \right)$$

It's pretty good at doing sums.

```
> sum(((1+k)!)^2/(1+2*k)!*x^k, k=0..infinity);
```

$$\frac{3}{4 \left(\frac{1}{4} x - 1\right)^2} + \frac{1}{2} \frac{\left(-1 - \frac{1}{2} x\right) \sqrt{1 - \frac{1}{4} x}\, \arcsin\left(\frac{1}{2} \sqrt{x}\right)}{\left(\frac{1}{4} x - 1\right)^3 \sqrt{x}}$$

```
> sum((2*k+1)^2/(k!)^2*x^k,k=0..infinity);
```

$$\mathrm{BesselI}(0, 2\sqrt{x}) + 4\sqrt{x}\, \mathrm{BesselI}(1, 2\sqrt{x}) + 4 x\, \mathrm{BesselI}(0, 2\sqrt{x})$$

Of course, sometimes there's no "closed form" formula.

```
> sum(x^k/(1+2^k), k=0..infinity);
```

$$\sum_{k=0}^{\infty} \frac{x^k}{1 + 2^k}$$

But suppose you only know a finite number of terms of the series. Is there any hope? Theoretically, no: the series could continue in all sorts of ways, e.g. the coefficients might all be 0 from this point on. But Maple might be able to "guess" how it continues. The appropriate function is **guessgf** in the **gfun** package. Here's a list of numbers.

```
> L := [1/2,1/4,1/6,1/8,1/10];
```

$$L := \left[\frac{1}{2}, \frac{1}{4}, \frac{1}{6}, \frac{1}{8}, \frac{1}{10}\right]$$

What's a likely function whose Maclaurin series is $L_1 + L_2\,x + L_3\,x^2 + ...$?

```
> with(gfun):
> guessgf(L,x,[ogf]);
```

$$\left[-\frac{1}{2}\,\frac{\ln(1-x)}{x}, ogf\right]$$

```
> taylor(%[1],x,20);
```

$$\frac{1}{2} + \frac{1}{4}\,x + \frac{1}{6}\,x^2 + \frac{1}{8}\,x^3 + \frac{1}{10}\,x^4 + \frac{1}{12}\,x^5 + \frac{1}{14}\,x^6 + \frac{1}{16}\,x^7 + \frac{1}{18}\,x^8 + \frac{1}{20}\,x^9 + \frac{1}{22}\,x^{10} + \frac{1}{24}\,x^{11} + \frac{1}{26}\,x^{12} + \frac{1}{28}\,x^{13} + \frac{1}{30}\,x^{14} + \frac{1}{32}\,x^{15} + \frac{1}{34}\,x^{16} + \frac{1}{36}\,x^{17} + \frac{1}{38}\,x^{18} + \mathrm{O}(x^{19})$$

```
> convert(%%[1],FormalPowerSeries,x);
```

$$\sum_{k=0}^{\infty} \frac{x^k}{2\,k+2} \qquad \textbf{(2.1)}$$

That was easy. Here's one that's not quite so obvious.

```
> guessgf([1,2,4,7,11,16,22],x,[ogf]);
```

$$\left[\frac{-1+x-x^2}{(x-1)^3}, ogf\right]$$

```
> taylor(%[1],x,8);
```

$$1 + 2\,x + 4\,x^2 + 7\,x^3 + 11\,x^4 + 16\,x^5 + 22\,x^6 + 29\,x^7 + \mathrm{O}(x^8)$$

```
> convert(%%[1],FormalPowerSeries,x);
```

$$\sum_{k=0}^{\infty} \left(1 + \frac{1}{2}\,k + \frac{1}{2}\,k^2\right) x^k \qquad \textbf{(2.2)}$$

The **ogf** stands for "ordinary generating function". A function $f(x)$ is the ordinary generating function of the sequence $c_0, c_1, c_2, ...$ if that sequence is the sequence of Maclaurin series coefficients of $f(x)$, i.e. $f(x) = \sum_{k=0}^{\infty} c_k x^k$

There's also **egf** or "**exponential generating function**", for a function whose coefficients are $\frac{L_1}{0!}$, $\frac{L_2}{1!}$, $\frac{L_3}{2!}$, ... That's not as useful for us here.

Also in the package is **listtoalgeq**, which would find a polynomial equation in x and y satisfied when y is the series with coefficients given by the list. It wouldn't work for our first implicit example, because the equation there involved exponentials. But try this one:

$x^2 y - x y^3 + y - 1 = 0$ with $y(0) = 1$.

```
> f := (x,y) -> x^2*y - x*y^3 + y - 1;
```

$$f := (x, y) \rightarrow x^2 y - x y^3 + y - 1$$

Note that $f(0, 1) = 0$. I'll use the Newton's method trick to find the Taylor series of the solution $y(x)$ about $x = 0$, then see if **listtoalgeq** will find the equation that $y(x)$ satisfies.

```
> newt:= (y,n) -> convert(normal(taylor(y-f(x,y)/D[2](f)(x,y),
  x=0, n)),polynom);
```

$$newt := (y, n) \rightarrow convert\left(normal\left(taylor\left(y - \frac{f(x, y)}{D_2(f)(x, y)}, x = 0, n\right)\right), polynom\right)$$

```
> Y[0] := 1;
  for j from 1 to 5 do
    Y[j] := newt(Y[j-1],2^j)
  end do;
```

$$Y_0 := 1$$

$$Y_1 := 1 + x$$

$$Y_2 := 1 + x + 2 x^2 + 8 x^3$$

$$Y_3 := 1 + x + 2 x^2 + 8 x^3 + 35 x^4 + 163 x^5 + 796 x^6 + 4024 x^7$$

$$Y_4 := 1 + x + 2 x^2 + 8 x^3 + 35 x^4 + 163 x^5 + 796 x^6 + 4024 x^7 + 20885 x^8 + 110654 x^9 + 596064 x^{10} + 3254752 x^{11} + 17974893 x^{12} + 100227022 x^{13} + 563482140 x^{14} + 3190633232 x^{15}$$

$$Y_5 := 1 + x + 2 x^2 + 8 x^3 + 35 x^4 + 163 x^5 + 796 x^6 + 4024 x^7 + 20885 x^8 + 110654 x^9 + 596064 x^{10} + 3254752 x^{11} + 17974893 x^{12} + 100227022 x^{13} + 563482140 x^{14} + 3190633232 x^{15} + 18179765509 x^{16} + 104158703503 x^{17} + 599698459613 x^{18} + 3467978715612 x^{19} + 20134256546896 x^{20} + 117313279477959 x^{21} + 685756774642494 x^{22} + 4020515276730588 x^{23} + 23636036336651811 x^{24} + 139301059260764048 x^{25} + 822881759633309667 x^{26} + 4871350637075703196 x^{27} + 28895082181969536230 x^{28} + 171712367070082813220 x^{29} + 1022183276503900838428 x^{30} + 6094767743827565180092 x^{31}$$

Can Maple take the list of coefficients and get the equation?

```
> L := [seq(coeff(Y[5],x,j),j=0..31)];
```

$L :=$ [1, 1, 2, 8, 35, 163, 796, 4024, 20885, 110654, 596064, 3254752, 17974893, 100227022, 563482140, 3190633232, 18179765509, 104158703503, 599698459613, 3467978715612, 20134256546896, 117313279477959, 685756774642494, 4020515276730588, 23636036336651811, 139301059260764048, 822881759633309667, 4871350637075703196, 28895082181969536230, 171712367070082813220, 1022183276503900838428, 6094767743827565180092]

```
> listtoalgeq(L,y(x));
```

$$[1 + (-1 - x^2)\, y(x) + x\, y(x)^3, ogf]$$

The first element here is equivalent to our $f(x, y(x))$ (well, actually it's $-f(x, y(x))$).

```
> %[1] + f(x,y(x));
```

$$(-1 - x^2)\, y(x) + x^2\, y(x) + y(x)$$

```
> expand(%);
```

$$0 \qquad \textbf{(2.3)}$$

```
> convert(RootOf(f(x,y),y),FormalPowerSeries,x);
```

$$RootOf\left(1 + x\,\_Z^3 + (-1 - x^2)\,\_Z\right) \qquad \textbf{(2.4)}$$

By the way, there's another interesting method of identifying a sequence of integers: the Encyclopedia of Integer Sequences. For example, what about the sequence 1, 1, 3, 15, 108, 1032, 12388, ...?

Look it up at http://www.research.att.com/~njas/sequences/index.html

A functional equation

The Encyclopedia of Integer Sequences says the ordinary generating function of sequence A090351 (i.e. the function whose Maclaurin series is that sequence) satisfies the equation

$$A(x)^3 = \frac{A\left(\frac{x}{1-x}\right)^2}{1-x}$$

That's called a functional equation: an equation involving values of an unknown function at different points. Can we use Maple to solve it, recovering the sequence?

```
> eq:= A(x)^3 = A(x/(1-x))^2/(1-x);
```

$$eq := A(x)^3 = \frac{A\left(\frac{x}{1-x}\right)^2}{1-x}$$

Note that for $x = 0$ we get

```
> eval(eq, x=0);
```

$$A(0)^3 = A(0)^2$$

so A(0) = 0 or 1. The one we want is 1.

```
> Aseries := unapply(1 + add(a[j]*x^j, j=1..20), x);
```

$$Aseries := x \to 1 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 + a_8 x^8 + a_9 x^9 + a_{10} x^{10} + a_{11} x^{11} + a_{12} x^{12} + a_{13} x^{13} + a_{14} x^{14} + a_{15} x^{15} + a_{16} x^{16} + a_{17} x^{17} + a_{18} x^{18} + a_{19} x^{19} + a_{20} x^{20}$$

The fact that $\frac{x}{1-x} = 0$ when $x = 0$ makes it possible to substitute this series in to the equation.

```
> eval(eq, A = Aseries);
```

$$\left(1 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 + a_6 x^6 + a_7 x^7 + a_8 x^8 + a_9 x^9 + a_{10} x^{10} + a_{11} x^{11} + a_{12} x^{12} + a_{13} x^{13} + a_{14} x^{14} + a_{15} x^{15} + a_{16} x^{16} + a_{17} x^{17} + a_{18} x^{18} + a_{19} x^{19} + a_{20} x^{20}\right)^3 = \frac{1}{1-x}\left(1 + \frac{a_1 x}{1-x} + \frac{a_2 x^2}{(1-x)^2} + \frac{a_3 x^3}{(1-x)^3} + \frac{a_4 x^4}{(1-x)^4}\right.$$

$$+\frac{a_5 x^5}{(1-x)^5}+\frac{a_6 x^6}{(1-x)^6}+\frac{a_7 x^7}{(1-x)^7}+\frac{a_8 x^8}{(1-x)^8}+\frac{a_9 x^9}{(1-x)^9}+\frac{a_{10} x^{10}}{(1-x)^{10}}$$
$$+\frac{a_{11} x^{11}}{(1-x)^{11}}+\frac{a_{12} x^{12}}{(1-x)^{12}}+\frac{a_{13} x^{13}}{(1-x)^{13}}+\frac{a_{14} x^{14}}{(1-x)^{14}}+\frac{a_{15} x^{15}}{(1-x)^{15}}+\frac{a_{16} x^{16}}{(1-x)^{16}}$$
$$\left.+\frac{a_{17} x^{17}}{(1-x)^{17}}+\frac{a_{18} x^{18}}{(1-x)^{18}}+\frac{a_{19} x^{19}}{(1-x)^{19}}+\frac{a_{20} x^{20}}{(1-x)^{20}}\right)^2$$

In the PDF version of this lesson, I'll use a colon on the next two commands, so those who print them out don't waste a lot of paper and ink.

```
> taylor(lhs(%)-rhs(%),x,21):
```

```
> eqs:= {seq(coeff(%,x,j),j=0..20)}:
```

```
> S:= solve(eqs);
```

$S := \{a_1 = 1, a_2 = 3, a_3 = 15, a_4 = 108, a_5 = 1032, a_6 = 12388, a_7 = 179572, a_8 = 3052986, a_9 = 59555338, a_{10} = 1310677726, a_{11} = 32114051862, a_{12} = 866766965308, a_{13} = 25547102523604, a_{14} = 816335926158372, a_{15} = 28107705687291892, a_{16} = 1037367351120788551, a_{17} = 40852168787823027351, a_{18} = 1709792654612819858341, a_{19} = 75786181910268208068217, a_{20} = 3546463856783571869500968\}$

These numbers get large rather quickly. It's not at all obvious that the radius of convergence of the series would be positive.

```
> [seq([n,evalf(eval(a[n],S)^(1/n))],n=1..19)];
```

[[1, 1.], [2, 1.732050808], [3, 2.466212074], [4, 3.223709795], [5, 4.006230560], [6, 4.810241337], [7, 5.631259190], [8, 6.465328582], [9, 7.309372199], [10, 8.161116771], [11, 9.018910982], [12, 9.881552808], [13, 10.74815630], [14, 11.61805675], [15, 12.49074550], [16, 13.36582547], [17, 14.24298081], [18, 15.12195597], [19, 16.00254101]]

```
> with(plots):
  pointplot(%%);
```



It looks like $\lim_{n \to \infty} a_n^{\left(\frac{1}{n}\right)} = \infty$, so the radius of convergence is 0.

Maple objects introduced in this lesson

gfun package
guessgf (in **gfun**)
listtoalgeq (in **gfun**)